

MinML Syntax

1 The MinML grammar

The concrete syntax of MinML is specified by the grammar given in Figures ?? and ??.

There are four classes of identifiers: *vid* for value identifiers (variables), *cid* for data constructor identifiers, *tyv* for type variables, and *tyc* for type constructors (including type constants). The identifier classes *vid* and *tyv* consist of alphanumeric identifiers starting with a lower-case letter, while *cid* and *tyc* range over alphanumeric identifier starting with an upper-case letter.

As written, this grammar is ambiguous. To make this grammar unambiguous, the precedence of operators must be specified. The precedence of the binary operators are (from weakest to strongest):

| |
& &
== <> < <= > >=
@
+ -
* / %

All binary operators are left associative except “@” (string concatenation) which is right associative. The next highest precedence is function application, which associates to the left. Here are some examples:

$$\begin{array}{lll} a + b * c + d & \equiv & (a + (b * c)) + d \\ f\ a\ @\ b\ @\ " & \equiv & (f\ a)\ @\ (b\ @\ "") \\ hd\ l\ x\ y & \equiv & ((hd\ l)\ x)\ y \end{array}$$

The lexical structure of MinML is defined in the ML-Lex specification file `minml.lex`, and the Yacc-style grammar is defined in `minml.grm`.

$$\begin{aligned} \textit{Prog} & ::= (\textit{TopDecl} \ ; \)^* \textit{Exp} \\ \\ \textit{TopDecl} & ::= \textbf{type} \ \textit{tyc} \ \textit{TypeParams}^{opt} = \textit{Type} \\ & \quad | \quad \textbf{datatype} \ \textit{tyc} \ \textit{TypeParams}^{opt} = \textit{ConsDecl} \ (\ | \ \textit{ConsDecl})^* \\ & \quad | \quad \textit{ValueDecl} \\ \\ \textit{TypeParams} & ::= \textit{tyv} \\ & \quad | \quad (\ \textit{tyv} \ (\ , \ \textit{tyv})^* \) \\ \\ \textit{Type} & ::= \textit{Type} \rightarrow \textit{Type} \\ & \quad | \quad \textit{AtomicType} \ (\star \ \textit{AtomicType})^* \\ \\ \textit{AtomicType} & ::= \textit{tyv} \\ & \quad | \quad \textit{tyc} \ (\ (\ \textit{Type} \ (\ , \ \textit{Type})^* \))^{opt} \\ & \quad | \quad (\ \textit{Type} \) \\ \\ \textit{ConsDecl} & ::= \textit{cid} \ (\ (\ \textit{Type} \))^{opt} \\ \\ \textit{ValueDecl} & ::= \textbf{val} \ \textit{TuplePat} = \textit{Exp} \\ & \quad | \quad \textbf{fun} \ \textit{FunDef} \ (\textbf{and} \ \textit{FunDef})^* \\ \\ \textit{FunDef} & ::= \textbf{vid} \ \textit{TuplePat} = \textit{Exp} \end{aligned}$$

Figure 1: The concrete syntax of MinML (A)

```

Const
  ::= num
  | str
  | cid

Pat
  ::= Const
  | cid TuplePat
  | TuplePat

TuplePat
  ::= AtomicPat
  | ( AtomicPat ( , AtomicPat )* )

AtomicPat
  ::= vid
  | -

Match
  ::= Pat => Exp

Exp
  ::= vid
  | Const
  | Exp | | Exp
  | Exp && Exp
  | Exp == Exp
  | Exp <> Exp
  | Exp < Exp
  | Exp <= Exp
  | Exp > Exp
  | Exp >= Exp
  | Exp @ Exp
  | Exp + Exp
  | Exp - Exp
  | Exp * Exp
  | Exp / Exp
  | Exp % Exp
  | ~ Exp
  | Exp Exp
  | ( Exp ( , Exp )* )
  | ( Exp ( ; Exp )* )
  | if Exp then Exp else Exp
  | let ValueDecl+ in Exp ( ; Exp )* end
  | case Exp of Match ( | Match )* end
  | fn vid => Exp

```

Figure 2: The concrete syntax of MinML (B)